

Optimizing DiffEq Code

Chris Rackauckas

December 29, 2019

In this notebook we will walk through some of the main tools for optimizing your code in order to efficiently solve DifferentialEquations.jl. User-side optimizations are important because, for sufficiently difficult problems, most of the time will be spent inside of your `f` function, the function you are trying to solve. "Efficient" integrators are those that reduce the required number of `f` calls to hit the error tolerance. The main ideas for optimizing your DiffEq code, or any Julia function, are the following:

- Make it non-allocating
- Use StaticArrays for small arrays
- Use broadcast fusion
- Make it type-stable
- Reduce redundant calculations
- Make use of BLAS calls
- Optimize algorithm choice

We'll discuss these strategies in the context of small and large systems. Let's start with small systems.

0.1 Optimizing Small Systems (<100 DEs)

Let's take the classic Lorenz system from before. Let's start by naively writing the system in its out-of-place form:

```
function lorenz(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    [dx,dy,dz]
end
```

```
lorenz (generic function with 1 method)
```

Here, `lorenz` returns an object, `[dx,dy,dz]`, which is created within the body of `lorenz`.

This is a common code pattern from high-level languages like MATLAB, SciPy, or R's `deSolve`. However, the issue with this form is that it allocates a vector, `[dx,dy,dz]`, at each step. Let's benchmark the solution process with this choice of function:

```
using DifferentialEquations, BenchmarkTools
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 10.83 MiB
  allocs estimate: 101447
  -----
  minimum time:      3.461 ms (0.00% GC)
  median time:       8.035 ms (0.00% GC)
  mean time:         8.451 ms (20.22% GC)
  maximum time:      49.114 ms (83.09% GC)
  -----
  samples:           592
  evals/sample:      1
```

The BenchmarkTools package's `@benchmark` runs the code multiple times to get an accurate measurement. The minimum time is the time it takes when your OS and other background processes aren't getting in the way. Notice that in this case it takes about 5ms to solve and allocates around 11.11 MiB. However, if we were to use this inside of a real user code we'd see a lot of time spent doing garbage collection (GC) to clean up all of the arrays we made. Even if we turn off saving we have these allocations.

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 9.47 MiB
  allocs estimate: 88665
  -----
  minimum time:      3.021 ms (0.00% GC)
  median time:       7.002 ms (0.00% GC)
  mean time:         7.429 ms (20.44% GC)
  maximum time:      47.625 ms (85.67% GC)
  -----
  samples:           673
  evals/sample:      1
```

The problem of course is that arrays are created every time our derivative function is called. This function is called multiple times per step and is thus the main source of memory usage. To fix this, we can use the in-place form to *****make our code non-allocating*****:

```
function lorenz!(du,u,p,t)
  du[1] = 10.0*(u[2]-u[1])
  du[2] = u[1]*(28.0-u[3]) - u[2]
  du[3] = u[1]*u[2] - (8/3)*u[3]
end

lorenz! (generic function with 1 method)
```

Here, instead of creating an array each time, we utilized the cache array `du`. When the inplace form is used, `DifferentialEquations.jl` takes a different internal route that minimizes

the internal allocations as well. When we benchmark this function, we will see quite a difference.

```
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate:  1.38 MiB
  allocs estimate: 13009
  -----
  minimum time:     717.800 μs (0.00% GC)
  median time:      1.300 ms (0.00% GC)
  mean time:        1.372 ms (14.66% GC)
  maximum time:     39.204 ms (96.31% GC)
  -----
  samples:          3637
  evals/sample:     1
```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  6.28 KiB
  allocs estimate:  67
  -----
  minimum time:     311.301 μs (0.00% GC)
  median time:      362.400 μs (0.00% GC)
  mean time:        355.195 μs (0.00% GC)
  maximum time:     702.100 μs (0.00% GC)
  -----
  samples:          10000
  evals/sample:     1
```

There is a 4x time difference just from that change! Notice there are still some allocations and this is due to the construction of the integration cache. But this doesn't scale with the problem size:

```
tspan = (0.0,500.0) # 5x longer than before
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  6.28 KiB
  allocs estimate:  67
  -----
  minimum time:     1.553 ms (0.00% GC)
  median time:      1.832 ms (0.00% GC)
  mean time:        1.776 ms (0.00% GC)
  maximum time:     2.457 ms (0.00% GC)
  -----
  samples:          2813
  evals/sample:     1
```

since that's all just setup allocations.

But if the system is small we can optimize even more. Allocations are only expensive if they are "heap allocations". For a more in-depth definition of heap allocations, [there are](#)

a lot of sources online. But a good working definition is that heap allocations are variable-sized slabs of memory which have to be pointed to, and this pointer indirection costs time. Additionally, the heap has to be managed and the garbage controllers has to actively keep track of what's on the heap.

However, there's an alternative to heap allocations, known as stack allocations. The stack is statically-sized (known at compile time) and thus its accesses are quick. Additionally, the exact block of memory is known in advance by the compiler, and thus re-using the memory is cheap. This means that allocating on the stack has essentially no cost!

Arrays have to be heap allocated because their size (and thus the amount of memory they take up) is determined at runtime. But there are structures in Julia which are stack-allocated. `structs` for example are stack-allocated "value-type"s. `Tuples` are a stack-allocated collection. The most useful data structure for `DiffEq` though is the `StaticArray` from the package [StaticArrays.jl](#). These arrays have their length determined at compile-time. They are created using macros attached to normal array expressions, for example:

```
using StaticArrays
A = @SVector [2.0,3.0,5.0]

3-element StaticArrays.SArray{Tuple{3},Float64,1,3} with indices SOneTo(3):
 2.0
 3.0
 5.0
```

Notice that the 3 after `SVector` gives the size of the `SVector`. It cannot be changed. Additionally, `SVectors` are immutable, so we have to create a new `SVector` to change values. But remember, we don't have to worry about allocations because this data structure is stack-allocated. `SArrays` have a lot of extra optimizations as well: they have fast matrix multiplication, fast QR factorizations, etc. which directly make use of the information about the size of the array. Thus, when possible they should be used.

Unfortunately static arrays can only be used for sufficiently small arrays. After a certain size, they are forced to heap allocate after some instructions and their compile time balloons. Thus static arrays shouldn't be used if your system has more than 100 variables. Additionally, only the native Julia algorithms can fully utilize static arrays.

Let's `***optimize lorenz using static arrays***`. Note that in this case, we want to use the out-of-place allocating form, but this time we want to output a static array:

```
function lorenz_static(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    @SVector [dx,dy,dz]
end

lorenz_static (generic function with 1 method)
```

To make the solver internally use static arrays, we simply give it a static array as the initial condition:

```
u0 = @SVector [1.0,0.0,0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz_static,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```

BenchmarkTools.Trial:
  memory estimate:  467.31 KiB
  allocs estimate:  2608
  -----
  minimum time:     305.400 μs (0.00% GC)
  median time:      475.300 μs (0.00% GC)
  mean time:        512.149 μs (11.66% GC)
  maximum time:     34.178 ms (98.54% GC)
  -----
  samples:          9719
  evals/sample:     1

@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  4.36 KiB
  allocs estimate:  41
  -----
  minimum time:     204.999 μs (0.00% GC)
  median time:      241.301 μs (0.00% GC)
  mean time:        243.370 μs (0.00% GC)
  maximum time:     504.200 μs (0.00% GC)
  -----
  samples:          10000
  evals/sample:     1

```

And that's pretty much all there is to it. With static arrays you don't have to worry about allocating, so use operations like `*` and don't worry about fusing operations (discussed in the next section). Do "the vectorized code" of R/MATLAB/Python and your code in this case will be fast, or directly use the numbers/values.

Exercise 1 Implement the out-of-place array, in-place array, and out-of-place static array forms for the [Henon-Heiles System](#) and time the results.

0.2 Optimizing Large Systems

0.2.1 Interlude: Managing Allocations with Broadcast Fusion

When your system is sufficiently large, or you have to make use of a non-native Julia algorithm, you have to make use of `Arrays`. In order to use arrays in the most efficient manner, you need to be careful about temporary allocations. Vectorized calculations naturally have plenty of temporary array allocations. This is because a vectorized calculation outputs a vector. Thus:

```

A = rand(1000,1000); B = rand(1000,1000); C = rand(1000,1000)
test(A,B,C) = A + B + C
@benchmark test(A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:     3.636 ms (0.00% GC)
  median time:      4.187 ms (0.00% GC)
  mean time:        5.077 ms (17.24% GC)

```

```

maximum time:      36.399 ms (86.45% GC)
-----
samples:           983
evals/sample:      1

```

That expression `A + B + C` creates 2 arrays. It first creates one for the output of `A + B`, then uses that result array to `+ C` to get the final result. 2 arrays! We don't want that! The first thing to do to fix this is to use broadcast fusion. [Broadcast fusion](#) puts expressions together. For example, instead of doing the `+` operations separately, if we were to add them all at the same time, then we would only have a single array that's created. For example:

```

test2(A,B,C) = map((a,b,c)->a+b+c,A,B,C)
@benchmark test2(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  5
-----
 minimum time:     4.430 ms (0.00% GC)
 median time:     5.151 ms (0.00% GC)
 mean time:       5.965 ms (14.55% GC)
 maximum time:    37.420 ms (85.81% GC)
-----
 samples:          837
 evals/sample:     1

```

Puts the whole expression into a single function call, and thus only one array is required to store output. This is the same as writing the loop:

```

function test3(A,B,C)
    D = similar(A)
    @inbounds for i in eachindex(A)
        D[i] = A[i] + B[i] + C[i]
    end
    D
end
@benchmark test3(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
 minimum time:     3.600 ms (0.00% GC)
 median time:     4.150 ms (0.00% GC)
 mean time:       5.046 ms (17.56% GC)
 maximum time:    35.301 ms (87.83% GC)
-----
 samples:          989
 evals/sample:     1

```

However, Julia's broadcast is syntactic sugar for this. If multiple expressions have a `.`, then it will put those vectorized operations together. Thus:

```

test4(A,B,C) = A .+ B .+ C
@benchmark test4(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2

```

```

-----
minimum time:      3.643 ms (0.00% GC)
median time:       4.177 ms (0.00% GC)
mean time:         5.067 ms (17.62% GC)
maximum time:      37.125 ms (88.44% GC)
-----

samples:           985
evals/sample:      1

```

is a version with only 1 array created (the output). Note that `.s` can be used with function calls as well:

```
sin.(A) .+ sin.(B)
```

```

1000×1000 Array{Float64,2}:
 1.50216  0.965776  1.26746  0.832386  ...  1.16465  0.589113  1.34856
 1.33182  0.51365  1.37389  1.43623  ...  0.980262  0.779842  1.13034
 1.25698  0.471244  0.851849  0.618252  ...  0.775696  0.597301  0.909595
 0.870405 0.878794  0.685954  1.41097  ...  0.1714  0.680644  1.28561
 0.468142 1.04088  0.611257  0.893978  ...  0.188891  0.262612  0.292855
 1.0118  1.20311  0.910531  1.00178  ...  1.17343  0.776839  1.00285
 0.470712 1.18301  0.271359  0.82657  ...  1.19597  0.930059  0.853555
 0.564465 0.97572  1.32917  1.565  ...  0.448208  1.18904  0.522671
 1.27614  0.220454 1.01127  0.964145  ...  0.82187  0.87618  0.442406
 0.768597 0.748832 1.22524  0.782177  ...  0.579029  1.08053  0.588279
 ⋮
 1.12686  1.09869  0.97548  0.797588  ...  0.82027  1.44322  0.987908
 1.23789  1.24997  0.861792  0.815279  ...  1.32313  0.818792  1.13386
 0.738894 0.923651 0.368482 1.03204  ...  0.520827  0.912963  0.923039
 1.22473  0.860693 0.371965 0.106558  ...  0.970782  0.350235  1.0632
 1.05873  0.875997 0.835037 0.554735  ...  1.45532  1.22159  0.873442
 1.00606  1.24222  0.249463 1.14187  ...  1.09575  1.45183  1.18242
 1.36957  1.02324  1.51197  0.112333  ...  0.356421  0.509599  1.13946
 0.715417 0.643262 0.938109 0.517242  ...  0.994875  0.435153  0.940859
 0.59704  0.337149 0.449464 1.05077  ...  0.367627  0.872794  0.631273

```

Also, the `@.` macro applies a dot to every operator:

```
test5(A,B,C) = @. A + B + C #only one array allocated
@benchmark test5(A,B,C)
```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:      3.643 ms (0.00% GC)
median time:       4.389 ms (0.00% GC)
mean time:         5.340 ms (16.92% GC)
maximum time:      38.695 ms (87.01% GC)
-----

samples:           934
evals/sample:      1

```

Using these tools we can get rid of our intermediate array allocations for many vectorized function calls. But we are still allocating the output array. To get rid of that allocation, we can instead use mutation. Mutating broadcast is done via `.=`. For example, if we pre-allocate the output:

```
D = zeros(1000,1000);
```

Then we can keep re-using this cache for subsequent calculations. The mutating broadcasting form is:

```
test6!(D,A,B,C) = D .= A .+ B .+ C #only one array allocated
@benchmark test6!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:     1.508 ms (0.00% GC)
  median time:      1.656 ms (0.00% GC)
  mean time:        1.675 ms (0.00% GC)
  maximum time:     2.765 ms (0.00% GC)
  -----
  samples:          2963
  evals/sample:     1
```

If we use `@.` before the `=`, then it will turn it into `.=`:

```
test7!(D,A,B,C) = @. D = A + B + C #only one array allocated
@benchmark test7!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:     1.534 ms (0.00% GC)
  median time:      1.658 ms (0.00% GC)
  mean time:        1.674 ms (0.00% GC)
  maximum time:     2.424 ms (0.00% GC)
  -----
  samples:          2966
  evals/sample:     1
```

Notice that in this case, there is no "output", and instead the values inside of `D` are what are changed (like with the `DiffEq` inplace function). Many Julia functions have a mutating form which is denoted with a `!`. For example, the mutating form of the `map` is `map!`:

```
test8!(D,A,B,C) = map!((a,b,c)->a+b+c,D,A,B,C)
@benchmark test8!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  32 bytes
  allocs estimate:  1
  -----
  minimum time:     2.698 ms (0.00% GC)
  median time:      2.827 ms (0.00% GC)
  mean time:        2.887 ms (0.00% GC)
  maximum time:     5.117 ms (0.00% GC)
  -----
  samples:          1726
  evals/sample:     1
```

Some operations require using an alternate mutating form in order to be fast. For example, matrix multiplication via `*` allocates a temporary:

```
@benchmark A*B
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:      17.071 ms (0.00% GC)
  median time:       27.382 ms (0.00% GC)
  mean time:         28.172 ms (4.64% GC)
  maximum time:      80.259 ms (59.39% GC)
  -----
  samples:           178
  evals/sample:      1
```

Instead, we can use the mutating form `mul!` into a cache array to avoid allocating the output:

```
using LinearAlgebra
@benchmark mul!(D,A,B) # same as D = A * B
```

```
BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:      14.562 ms (0.00% GC)
  median time:       24.828 ms (0.00% GC)
  mean time:         25.266 ms (0.00% GC)
  maximum time:      109.913 ms (0.00% GC)
  -----
  samples:           198
  evals/sample:      1
```

For repeated calculations this reduced allocation can stop GC cycles and thus lead to more efficient code. Additionally, *******we can fuse together higher level linear algebra operations using BLAS*******. The package [SugarBLAS.jl](#) makes it easy to write higher level operations like `alpha*B*A + beta*C` as mutating BLAS calls.

0.2.2 Example Optimization: Gierer-Meinhardt Reaction-Diffusion PDE Discretization

Let's optimize the solution of a Reaction-Diffusion PDE's discretization. In its discretized form, this is the ODE:

$$du = D_1(A_y u + u A_x) + \frac{au^2}{v} + \bar{u} - \alpha u \quad (1)$$

$$dv = D_2(A_y v + v A_x) + au^2 + \beta v \quad (2)$$

where u , v , and A are matrices. Here, we will use the simplified version where A is the tridiagonal stencil $[1, -2, 1]$, i.e. it's the 2D discretization of the Laplacian. The native code would be something along the lines of:

```
# Generate the constants
p = (1.0,1.0,1.0,10.0,0.001,100.0) # a,α,ubar,β,D1,D2
N = 100
Ax = Array{Tridiagonal{Float64}}{1,N-1,N-1}([1.0 for i in 1:N-1],[-2.0 for i in 1:N],[1.0 for i in 1:N-1]))
Ay = copy(Ax)
Ax[2,1] = 2.0
```

```

Ax[end-1,end] = 2.0
Ay[1,2] = 2.0
Ay[end,end-1] = 2.0

function basic_version!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = r[:, :, 1]
    v = r[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    dr[:, :, 1] = Du .+ a.*u.*u./v .+ ubar .- α*u
    dr[:, :, 2] = Dv .+ a.*u.*u .- β*v
end

a,α,ubar,β,D1,D2 = p
uss = (ubar+β)/α
vss = (a/β)*uss^2
r0 = zeros(100,100,2)
r0[:, :, 1] .= uss.+0.1.*rand.()
r0[:, :, 2] .= vss

prob = ODEProblem(basic_version!,r0,(0.0,0.1),p)

ODEProblem with uType Array{Float64,3} and tType Float64. In-place: true
timespan: (0.0, 0.1)
u0: [11.01862706666514 11.050758078406671 ... 11.05481936578792 11.0539506205
3214; 11.016409830887056 11.04034359927839 ... 11.00262549221627 11.081133859
491226; ... ; 11.030318968121122 11.05121007031332 ... 11.05567363828492 11.018
140413290922; 11.063257359858836 11.06814887459553 ... 11.04027672069419 11.0
76033271918547]

[12.100000000000001 12.100000000000001 ... 12.100000000000001 12.100000000000
001; 12.100000000000001 12.100000000000001 ... 12.100000000000001 12.1000000
0000001; ... ; 12.100000000000001 12.100000000000001 ... 12.100000000000001 12.
1000000000000001; 12.100000000000001 12.100000000000001 ... 12.100000000000001
12.100000000000001]

```

In this version we have encoded our initial condition to be a 3-dimensional array, with $u[:, :, 1]$ being the A part and $u[:, :, 2]$ being the B part.

```
@benchmark solve(prob,Tsit5())
```

```

BenchmarkTools.Trial:
 memory estimate: 186.88 MiB
 allocs estimate: 8564
-----
 minimum time:      87.099 ms (0.00% GC)
 median time:      140.837 ms (19.23% GC)
 mean time:        139.294 ms (15.29% GC)
 maximum time:     237.780 ms (22.49% GC)
-----
 samples:           36
 evals/sample:      1

```

While this version isn't very efficient,

We recommend writing the "high-level" code first, and iteratively optimizing it! The first thing that we can do is get rid of the slicing allocations. The operation $r[:, :, 1]$

creates a temporary array instead of a "view", i.e. a pointer to the already existing memory. To make it a view, add `@view`. Note that we have to be careful with views because they point to the same memory, and thus changing a view changes the original values:

```
A = rand(4)
@show A

A = [0.48694635196805125, 0.6828419367703846, 0.5260910293571688, 0.3520494
711121265]

B = @view A[1:3]
B[2] = 2
@show A

A = [0.48694635196805125, 2.0, 0.5260910293571688, 0.3520494711121265]
4-element Array{Float64,1}:
 0.48694635196805125
 2.0
 0.5260910293571688
 0.3520494711121265
```

Notice that changing B changed A. This is something to be careful of, but at the same time we want to use this since we want to modify the output `dr`. Additionally, the last statement is a purely element-wise operation, and thus we can make use of broadcast fusion there. Let's rewrite `basic_version!` to `***avoid slicing allocations***` and to `***use broadcast fusion***`:

```
function gm2!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    @. du = Du + a.*u.*u./v + ubar - α*u
    @. dv = Dv + a.*u.*u - β*v
end
prob = ODEProblem(gm2!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
 memory estimate: 119.55 MiB
  allocs estimate: 7094
  -----
 minimum time:      74.030 ms (0.00% GC)
 median time:      104.817 ms (0.00% GC)
 mean time:        108.263 ms (12.67% GC)
 maximum time:     162.376 ms (34.72% GC)
  -----
 samples:          47
 evals/sample:     1
```

Now, most of the allocations are taking place in `Du = D1*(Ay*u + u*Ax)` since those operations are vectorized and not mutating. We should instead replace the matrix multiplications with `mul!`. When doing so, we will need to have cache variables to write into. This looks like:

```

Ayu = zeros(N,N)
uAx = zeros(N,N)
Du = zeros(N,N)
Ayv = zeros(N,N)
vAx = zeros(N,N)
Dv = zeros(N,N)
function gm3!(dr,r,p,t)
    a, $\alpha$ ,ubar, $\beta$ ,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar -  $\alpha$ *u
    @. dv = Dv + a*u*u -  $\beta$ *v
end
prob = ODEProblem(gm3!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

BenchmarkTools.Trial:
 memory estimate: 29.76 MiB
  allocs estimate: 5330
  -----
 minimum time:      46.099 ms (0.00% GC)
 median time:      63.705 ms (0.00% GC)
 mean time:        64.865 ms (3.72% GC)
 maximum time:     99.879 ms (29.06% GC)
  -----
 samples:          78
 evals/sample:     1

```

But our temporary variables are global variables. We need to either declare the caches as `const` or localize them. We can localize them by adding them to the parameters, `p`. It's easier for the compiler to reason about local variables than global variables. *****Localizing variables helps to ensure type stability*****.

```

p = (1.0,1.0,1.0,10.0,0.001,100.0,Ayu,uAx,Du,Ayv,vAx,Dv) # a, $\alpha$ ,ubar, $\beta$ ,D1,D2
function gm4!(dr,r,p,t)
    a, $\alpha$ ,ubar, $\beta$ ,D1,D2,Ayu,uAx,Du,Ayv,vAx,Dv = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar -  $\alpha$ *u
    @. dv = Dv + a*u*u -  $\beta$ *v
end
prob = ODEProblem(gm4!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 29.66 MiB
  allocs estimate: 1066
  -----
  minimum time:      45.568 ms (0.00% GC)
  median time:       53.054 ms (0.00% GC)
  mean time:         55.902 ms (4.47% GC)
  maximum time:      91.089 ms (27.20% GC)
  -----
  samples:           90
  evals/sample:      1

```

We could then use the BLAS `gemmv` to optimize the matrix multiplications some more, but instead let's devectorize the stencil.

```

p = (1.0,1.0,1.0,10.0,0.001,100.0,N)
function fast_gm!(du,u,p,t)
  a,α,ubar,β,D1,D2,N = p

  @inbounds for j in 2:N-1, i in 2:N-1
    du[i,j,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
      a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
  end

  @inbounds for j in 2:N-1, i in 2:N-1
    du[i,j,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
      a*u[i,j,1]^2 - β*u[i,j,2]
  end

  @inbounds for j in 2:N-1
    i = 1
    du[1,j,1] = D1*(2u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
      a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
  end
  @inbounds for j in 2:N-1
    i = 1
    du[1,j,2] = D2*(2u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
      a*u[i,j,1]^2 - β*u[i,j,2]
  end

  @inbounds for j in 2:N-1
    i = N
    du[end,j,1] = D1*(2u[i-1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
      a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
  end
  @inbounds for j in 2:N-1
    i = N
    du[end,j,2] = D2*(2u[i-1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
      a*u[i,j,1]^2 - β*u[i,j,2]
  end

  @inbounds for i in 2:N-1
    j = 1
    du[i,1,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
      a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
  end
  @inbounds for i in 2:N-1
    j = 1
    du[i,1,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
      a*u[i,j,1]^2 - β*u[i,j,2]
  end
end

```

```

@inbounds for i in 2:N-1
    j = N
    du[i,end,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds begin
    i = 1; j = 1
    du[1,1,1] = D1*(2u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,1,2] = D2*(2u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = 1; j = N
    du[1,N,1] = D1*(2u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,N,2] = D2*(2u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = 1
    du[N,1,1] = D1*(2u[i-1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[N,1,2] = D2*(2u[i-1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = N
    du[end,end,1] = D1*(2u[i-1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[end,end,2] = D2*(2u[i-1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
end
prob = ODEProblem(fast_gm!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
 memory estimate: 29.63 MiB
 allocs estimate: 479
-----
 minimum time:      10.254 ms (0.00% GC)
 median time:       12.286 ms (0.00% GC)
 mean time:         14.812 ms (13.67% GC)
 maximum time:      60.218 ms (72.77% GC)
-----
 samples:           338
 evals/sample:      1

```

Lastly, we can do other things like multithread the main loops, but these optimizations get the last 2x-3x out. The main optimizations which apply everywhere are the ones we just performed (though the last one only works if your matrix is a stencil. This is known as a matrix-free implementation of the PDE discretization).

This gets us to about 8x faster than our original MATLAB/SciPy/R vectorized style code!

The last thing to do is then *****optimize our algorithm choice*****. We have been using `Tsit5()` as our test algorithm, but in reality this problem is a stiff PDE discretization and thus one recommendation is to use `CVODE_BDF()`. However, instead of using the default dense Jacobian, we should make use of the sparse Jacobian afforded by the problem. The Jacobian is the matrix $\frac{df_i}{dr_j}$, where r is read by the linear index (i.e. down columns). But since the u variables depend on the v , the band size here is large, and thus this will not do well with a Banded Jacobian solver. Instead, we utilize sparse Jacobian algorithms. `CVODE_BDF` allows us to use a sparse Newton-Krylov solver by setting `linear_solver = :GMRES` (see [the solver documentation](#), and thus we can solve this problem efficiently. Let's see how this scales as we increase the integration time.

```
prob = ODEProblem(fast_gm!, r0, (0.0, 10.0), p)
@benchmark solve(prob, Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate:  2.76 GiB
  allocs estimate:  41645
  -----
  minimum time:     2.578 s (27.36% GC)
  median time:      2.746 s (30.25% GC)
  mean time:        2.746 s (30.25% GC)
  maximum time:     2.913 s (32.82% GC)
  -----
  samples:          2
  evals/sample:     1
```

```
using Sundials
@benchmark solve(prob, CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate:  126.31 MiB
  allocs estimate:  21496
  -----
  minimum time:     650.473 ms (0.00% GC)
  median time:      759.254 ms (0.00% GC)
  mean time:        751.765 ms (1.48% GC)
  maximum time:     820.989 ms (5.24% GC)
  -----
  samples:          7
  evals/sample:     1
```

```
prob = ODEProblem(fast_gm!, r0, (0.0, 100.0), p)
# Will go out of memory if we don't turn off `save_everystep`!
@benchmark solve(prob, Tsit5(), save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  2.90 MiB
  allocs estimate:  87
  -----
  minimum time:     6.392 s (0.00% GC)
  median time:      6.392 s (0.00% GC)
  mean time:        6.392 s (0.00% GC)
  maximum time:     6.392 s (0.00% GC)
  -----
  samples:          1
  evals/sample:     1
```

```
@benchmark solve(prob, CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate: 340.98 MiB
  allocs estimate: 65156
  -----
  minimum time:      1.994 s (0.44% GC)
  median time:       2.183 s (0.45% GC)
  mean time:         2.129 s (1.03% GC)
  maximum time:      2.209 s (2.15% GC)
  -----
  samples:           3
  evals/sample:      1
```

Now let's check the allocation growth.

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 3.78 MiB
  allocs estimate: 49464
  -----
  minimum time:      1.892 s (0.00% GC)
  median time:       2.032 s (0.00% GC)
  mean time:         1.991 s (0.00% GC)
  maximum time:      2.050 s (0.00% GC)
  -----
  samples:           3
  evals/sample:      1
```

```
prob = ODEProblem(fast_gm!,r0,(0.0,500.0),p)
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 5.68 MiB
  allocs estimate: 82088
  -----
  minimum time:      3.217 s (0.00% GC)
  median time:       3.285 s (0.00% GC)
  mean time:         3.285 s (0.00% GC)
  maximum time:      3.353 s (0.00% GC)
  -----
  samples:           2
  evals/sample:      1
```

Notice that we've eliminated almost all allocations, allowing the code to grow without hitting garbage collection and slowing down.

Why is `CVODE_BDF` doing well? What's happening is that, because the problem is stiff, the number of steps required by the explicit Runge-Kutta method grows rapidly, whereas `CVODE_BDF` is taking large steps. Additionally, the `GMRES` linear solver form is quite an efficient way to solve the implicit system in this case. This is problem-dependent, and in many cases using a Krylov method effectively requires a preconditioner, so you need to play around with testing other algorithms and linear solvers to find out what works best with your problem.

0.3 Conclusion

Julia gives you the tools to optimize the solver "all the way", but you need to make use of it. The main thing to avoid is temporary allocations. For small systems, this is effectively done

via static arrays. For large systems, this is done via in-place operations and cache arrays. Either way, the resulting solution can be immensely sped up over vectorized formulations by using these principles.

0.4 Appendix

This tutorial is part of the DiffEqTutorials.jl repository, found at: <https://github.com/JuliaDiffEq/DiffEqTutorials>

To locally run this tutorial, do the following commands:

```
using DiffEqTutorials
DiffEqTutorials.weave_file("introduction","03-optimizing_diffeq_code.jmd")
```

Computer Information:

Julia Version 1.3.0

Commit 46ce4d7933 (2019-11-26 06:09 UTC)

Platform Info:

OS: Windows (x86_64-w64-mingw32)

CPU: Intel(R) Core(TM) i7-8550U CPU @ 1.80GHz

WORD_SIZE: 64

LIBM: libopenlibm

LLVM: libLLVM-6.0.1 (ORCJIT, skylake)

Environment:

JULIA_EDITOR = "C:\Users\accou\AppData\Local\atom\app-1.42.0\atom.exe" -a

JULIA_NUM_THREADS = 4

Package Information:

Status `~\.julia\dev\DiffEqTutorials\Project.toml`